The George Washington University
Department of Statistics

STAT 6197 – Spring 2019

Week 2 – January 25, 2019

Major Topic: Reading Data into SAS, and Creating Customized Reports Using DATA Step

Detailed Topics:

1. Reading Raw Data Files into SAS Using
    a. Column Input
    b. Formatted Input
    c. List Input
    d. Named Input
2. Reading Microsoft Excel Data into SAS
3. Handling Missing Values
4. Compilation vs. Execution
5. Creating Reports Using DATA Step

Readings:

1. Relevant Chapters/Sections - Delwiche L, and Slaughter S. *The Little SAS Book: A Primer*, Fifth Edition Paperback – November 7, 2012.

2. Exercises from Relevant Chapters/Sections - Ottesen RA, Delwiche LD, and Slaughter SJ. *Exercises and Projects for The Little SAS Book*, Fifth Edition Paperback – July 1, 2015.

3. Reading Raw Data with the INPUT Statement (SAS Documentation)
4. Some Rules for Reading Numeric and Character Data (SAS Documentation)
5. Ways to represent missing values for numeric variables in SAS
6. Call Missing Routine (SAS(R) 9.4 Documentation)
7. Six ways to list variables in SAS (SAS Blogs)
8. PUT it there! Six tips for using PUT and %PUT Statements in SAS (SAS Blogs)

Raw Data Files

A raw data file is also known as a *flat file*.

- They are text files that contain one record per line.
- A record typically contains multiple fields.
- Flat files do not have internal metadata.
- External documentation, known as a *record layout*, should exist.
- A record layout describes the fields and locations within each record.

3

Raw Data Files

Fields in a raw data file can be delimited or arranged in fixed columns.

Delimited File

```
120102,Tom,Zhou,M,108255,Sales Manager,AU,11AUG1973,06/01/1993
120103,Wilson,Dawes,M,87975,Sales Manager,AU,22JAN1953,01/01/1978
120121,Irenie,Elvish,F,26600,Sales Rep. II,AU,02AUG1948,01/01/1978
120122,Christina,Ngan,F,27475,Sales Rep. II,AU,27JUL1958,07/01/1982
```

Fixed Column File

```
          1    1    2    2    3    3    4    4    5    5    6
1---5----0----5----0----5----0----5----0----5----0----5----0--
120102Tom        Zhou           Sales Manager       108255AU
120103Wilson     Dawes          Sales Manager        87975AU
120121Irenie     Elvish         Sales Rep. II        26600AU
120122Christina  Ngan           Sales Rep. II        27475AU
```

4

Reading Raw Data Files

These are the steps for creating a SAS data set from a raw data file.

Step 1	Name the output SAS data set.
Step 2	Locate and name the input raw data file.
Step 3	Examine the raw data file.
Step 4	Examine the file layout.
Step 5	Give SAS names to the fields.
Step 6	Determine the variable type.

8

Data Types

A SAS data set supports two types of variables.

Character variables

- can contain any value: letters, numerals, special characters, and blanks
- range from 1 to 32,767 characters in length
- have 1 byte per character.

Numeric variables

- store numeric values using floating point or binary representation
- have 8 bytes of storage by default
- can store 16 or 17 significant digits.

9

Determining the Variable Type

These are the rules for ***character*** variable values.

A character variable can contain	Examples
letters (A - Z, a - z)	"Ms. Helen Jones"
numbers (0 - 9)	"389"
special characters (!, @, #, %, $, and so on)	"$21,756"
a combination of the above	"648 Pine St." "Research & Development"

20

Determining the Variable Type

These are the rules for ***numeric*** variable values.

A numeric variable can contain	Examples
numbers (0-9)	175 4856281
a decimal point (.)	29.92
a negative sign (–)	–30
a letter E to indicate scientific notation	3.1E6

21

Standard and Nonstandard Data (Review)

Standard data is data that SAS can read without any additional instruction.

- Character data is always standard.
- Some numeric values are standard and some are not.

Standard Numeric Data	Nonstandard Numeric Data
58	(23)
67.23	-23	$67.23
5.67E5	00.99	5,823	01/12/2010
1.2E-2	12May2009

8

The following are the only acceptable characters in a standard numeric field:

```
0 1 2 3 4 5 6 7 8 9 . E e D d - +
```

Leading or trailing blanks are also acceptable.

E, **e**, **D**, and **d** represent exponential notation in a standard numeric field. An alternate way of writing **300000**, for example, is **3E5**.

SAS Date Values

SAS stores calendar dates as numeric values.



A *SAS date value* is stored as the number of days between January 1, 1960, and a specific date.

11

Acknowledgments: Portions of SAS' copyrighted SAS course content are reproduced here with permission of SAS Institute Inc., Cary, NC, USA. SAS Institute Inc. makes no warranties with respect to these materials and disclaims all liability therefor. Parts of additional materials are adapted from SAS Documentation, and web-based resources. Please let me know of any errors or typos you have found. Do not copy or circulate the handouts.

Input Styles in SAS



Acknowledgments: Portions of SAS' copyrighted SAS course content are reproduced here with permission of SAS Institute Inc., Cary, NC, USA. SAS Institute Inc. makes no warranties with respect to these materials and disclaims all liability therefor. Parts of additional materials are adapted from SAS Documentation, and web-based resources. Please let me know of any errors or typos you have found. Do not copy or circulate the handouts.

Input Style, and Types of Data Values and Data Fields

Input Style	Data Value Type	Data Field-Type (Location: Instream or External File)
Column Input	• Standard numeric data values • Character data values	Fixed fields
Formatted Input	• Standard numeric data values • Nonstandard data values • Character data values	Fixed fields
Simple List Input	• Standard numeric data values • Character data values containing *no embedded blanks*	Not arranged in fixed fields (i.e., unaligned or free format)
Modified List Input	• Standard numeric data values • Nonstandard data values • Character data values including those with *embedded blanks*	Not arranged in fixed fields (i.e., unaligned or free format)
Named Input	• Standard numeric data values • Character data values	Both variables and their values in the data file

Modified List Input = List Input + some features of Formatted Input.



Reading Raw Data Files: Input Styles



Input Styles

Column input, formatted input, list input, and named input are all styles of writing INPUT statement specifications.

Style	Use for Reading
Column input	Standard data in fixed columns
Formatted input	Standard and nonstandard data in fixed columns
List input	Standard and nonstandard data separated by blanks or some other delimiter
Named input	Standard data that is preceded by the name of the variable and an equal sign (=)

7

The list input style that handles standard data and character variables whose values are of no more than 8 bytes long is called *simple list input*.

The list input style that additionally handles nonstandard data and character variables whose values are more than 8 bytes long is called *modified list input*.

Modified List Input = List input + some features of Formatted Input.

When to Use Column Input

Reading Fixed Column Files

Column input is used to read standard values from fixed column data files.

Partial **salesemp.dat**

```
          1    1    2    2    3    3    4    4    5    5    6
1---5----0----5----0----5----0----5----0----5----0----5----0--
120102 Tom        Zhou           Sales Manager       108255 AU
120103 Wilson     Dawes          Sales Manager       87975  AU
120121 Irenie     Elvish         Sales Rep. II       26600  AU
120122 Christina  Ngan           Sales Rep. II       27475  AU
120123 Kimiko     Hotstone       Sales Rep. I        26190  AU
```

- Formatted input is used to read standard and nonstandard values from fixed column data files.

16

Using Column Input

Column input requires a variable name, type, starting column, and ending column for each field to be read.

Partial **salesemp.dat**

```
          1    1    2    2    3    3    4    4    5    5    6
1---5----0----5----0----5----0----5----0----5----0----5----0--
120102 Tom        Zhou           Sales Manager       108255 AU
120103 Wilson     Dawes          Sales Manager       87975  AU
```

Name: EmpID
Type: Numeric
Columns: 1-7

Field	Type	Starting Column	Ending Column
Employee ID	Numeric	1	7
First Name	Character	8	17
Last Name	Numeric	19	32
Job Title	Character	34	52
Salary	Numeric	54	59
Country	Character	61	62

17



18 p2aad02



19

In the INFILE statement you can specify either the name the external file or point to the external file with a fileref defined with the FILENAME statement, unless you have instream data following the DATALINES statement. You may also need to specify one of more options to this statement.



Advantages of Column Input

- Data fields can be read in any order.
- The same field can be reread.
- Parts of a data field can be read
- Both leading and trailing blanks within the field are ignored.
- Character values can contain embedded blanks.
- Placeholders (e.g., period), are not required for missing data.

Disadvantage: The *Column Input* cannot read nonstandard data values.

GitHub: SASGateWay/SASCourse/Week2/Ex1_Column_Input.sas

When to Use Formatted Input

Records with standard numeric, nonstandard numeric or character data fields aligned in columns

Features of the formatted input style

- Starting position
- Variable
- INFORMAT

The informat name must contain a period (.) at the end for SAS to distinguish it (i.e., informat name) from a variable name as can be seen from the code below.



"Formatted input causes the pointer to move like that of column input to read a variable value. The pointer moves the length that is specified in the informat and stops at the next column." – SAS Documentation.

Formatted Input: Use of Absolute Pointer Control vs. Relative Pointer Control



Remember that when you use formatted input, the column pointer control moves to the first column following the field that was just read.

Acknowledgments: Portions of SAS' copyrighted SAS course content are reproduced here with permission of SAS Institute Inc., Cary, NC, USA. SAS Institute Inc. makes no warranties with respect to these materials and disclaims all liability therefor. Parts of additional materials are adapted from SAS Documentation, and web-based resources. Please let me know of any errors or typos you have found. Do not copy or circulate the handouts.

Informats

An *informat* is an instruction that SAS uses to read data values into a variable.



SAS uses the informat to determine the following:

- whether the variable is numeric or character
- the length of character variables

48

a

Informats

Raw Data Value	Informat	SAS Data Value
$12,345	COMMA7. DOLLAR7.	12345
$12.345	COMMAX7. DOLLARX7.	12345
€12.345	EUROX7.	12345
Australia	$11.	Australia
Australia	$CHAR11.	Australia
au	$UPCASE2.	AU
01/01/1960	MMDDYY10.	0
31/12/60	DDMMYY8.	365
31DEC1959	DATE9.	-1

53

The ANYDTDTE*w.* informat can be used to read data that has a variety of date forms.

Informats always contain a *w* value to indicate the width of the raw data field. A perod (.) ends the informat or separates the *w* value from the optional *d* value, which specifies the number of decimal places.

Detailed Description Selected of INFORMATs

Informat	Definition
$*w*.	Reads standard character data.
w.d	Reads standard numeric data.
COMMA*w.d* DOLLAR*w.d*	Reads nonstandard numeric data and removes embedded commas, blanks, dollar signs, percent signs, and hyphens.
COMMAX*w.d* DOLLARX*w.d*	Operates like COMMA*w.d* and DOLLAR*w.d*, but reverses the role of the decimal point and comma. This convention is common in many European countries.
EUROX*w.d*	Reads nonstandard numeric data and removes embedded characters in European currency.
PERCENT*w.d*	Operates like COMMA*w.d* but divides the number by 100 if it is followed by a percent sign (%).

- If the *d* is not specified, it defaults to 0. This means that COMMA7. and COMMA7.0 are equivalent.
- If the data contains decimal points, SAS ignores the *d* value and reads the number of decimal places that are actually in the input data. If the data does not contain decimal places, *d* specifies the number of decimal places to imply.
- DOLLAR*w.d* is an alias for COMMA*w.d*, and DOLLARX*w.d* is an alias for COMMAX*w.d*.
- More information about EUROX*w.d* is available through SAS Help and Documentation.

GitHub: SASGateWay/SASCourse/Week2/Ex4_Formatted_Input.sas
GitHub: SASGateWay/SASCourse/Week2/Ex5_formatted_column_input.sas
GitHub: SASGateWay/SASCourse/Week2/Ex14_Column_Formatted_Input.sas
GitHub: SASGateWay/SASCourse/Week2/Ex15_Absolute_Relative_Pointer_Controls.sas

GitHub: SASGateWay/SASCourse/Week2/Exa35_input_numeric_character_data.sas

SAS Format Examples

Selected SAS formats:

Format	Stored Value	Displayed Value
$4.	Programming	Prog
12.	27134.5864	27135
12.2	27134.5864	27134.59
COMMA12.2	27134.5864	27,134.59
DOLLAR12.2	27134.5864	$27,134.59
COMMAX12.2	27134.5864	27.134,59
EUROX12.2	27134.5864	€27.134,59

10

SAS Date Format Examples

SAS date formats display SAS date values in standard date forms.

Format	Stored Value	Displayed Value
MMDDYY10.	0	01/01/1960
MMDDYY8.	0	01/01/60
MMDDYY6.	0	010160
DDMMYY10.	365	31/12/1960
DDMMYY8.	365	31/12/60
DDMMYY6.	365	311260

13

LRECL= and PAD Options on the INFILE Statement (Handling Shorter Records)

Example data in an external file:

```
001 Tim Dyson    74 8724
002 Sam Larson   96 8224
003 Jane Miller  91
004 Bikas Das    90 8783
```

Shorter record (not padded with blanks)

With the PAD option on the INFILE statement, SAS pads the records that are read from an external file with blanks to the length that is specified in the LRECL= option or implied by the column position.

GitHub: SASGateway/SASCourse/Week2/Ex2_column_Input_PAD_Option.sas

TRUNCOVER Option on the INFILE Statement

Example data in an external file:

```
001 Tim Dyson    74 8724
002 Sam Larson   96  224
003 Jane Miller  91   24
004 Bikas Das    90    4
```

For the last three records (above), the very last field is shorter than expected.

The TRUNCOVER option on the INFILE statement correctly assign the contents of the input buffer to the last field.

GitHub: SASGateway/SASCourse/Week2/Ex3_column_Input_TRUNCOVER_Option.sas

When to Use List Input

- Free-Format Data (Space or Character-Delimited Files).
- Data fields have character or numeric values.

Reading Delimited Data Files (Review)

Characteristics of delimited data files:

- Data values are separated by a delimiting character.
- The delimiting character can be a blank, tab, comma, or any other character.

Example: Delimited Raw Data File



26

List Input

- You must specify the variables in the order that they appear in the raw data file, left to right.
- The default length for variables is 8 bytes.
- A space (blank) is the default delimiter.

pointer control	Moves the input pointer to a specified column in the input buffer.
variable	Names a variable that is assigned input values.
$	Indicates to store a variable value as a character value rather than as a numeric value.
:	Reads data values that need the additional instructions that informats can provide but are not aligned in columns.
informat	Specifies an informat to use to read the variable values.

58

Reading a Delimited Raw Data File

Use *INFILE* and *INPUT statements* in a DATA step to read a raw data file.

```
data work.subset;
   infile "&path\sales.csv" dlm=',';
   input Employee_ID First_Name $
         Last_Name $ Gender $ Salary
         Job_Title $ Country $;
run;
```

DATA *output-data-set*;
INFILE "*raw-data-file*" <**DLM**='*delimiter*'>;
INPUT *variable* <$> *variable* <$> … ;
RUN;

11 p108d01

INPUT Statement

The INPUT statement reads the data fields sequentially, left to right. Standard data fields require only a variable name and type.

Partial **sales.csv**

```
120102,Tom,Zhou,M,108255,Sales Manager,AU,11AUG1969,06/01/1989
120103,Wilson,Dawes,M,87975,Sales Manager,AU,22JAN1949,01/01/1974
120121,Irenie,Elvish,F,26600,Sales Rep. II,AU,02AUG1944,01/01/1974
```

```
input Employee_ID First_Name $ Last_Name $
      Gender $ Salary Job_Title $ Country $;
```

INPUT *variable* <$> *variable* <$> …;

- The optional dollar sign indicates a character variable.
- Default length for ***all*** variables is eight bytes, regardless of type.

13

GitHub: SASGateway/SASCourse/Week2/Ex7_Simple_List_Input.sas

When to Use :Modifier with an Informat in List Input

- Free-Format Data
- Data fields separated by 1+ blanks or other delimiters
- Data fields have standard numeric values or characters more than 8 bytes
- Data fields have non-standard numeric values



If your instream data include semicolons, you are required to use the DATALINES4 statement instead of the DATALINES statement and the use four semicolons (;;;) instead of one semicolon (;) to mark the end of instream data.

GitHub: SASGateway/SASCourse/Week2/Ex31_Datalines4.sas.



LENGTH Statement to Define the Length of Character variables

LENGTH Statement

The *LENGTH statement* defines the type and length of a variable.

LENGTH *variable(s)* **$** *length*;

```
data work.subset;
   length First_Name $ 12 Last_Name $ 18
          Gender $ 1 Job_Title $ 25
          Country $ 2;
   infile "&path\sales.csv" dlm=',';
   input Employee_ID First_Name $ Last_Name $
         Gender $ Salary Job_Title $ Country $;
run;
```

⚠ Put the LENGTH statement before the INPUT statement.

43 p108d02

Viewing the Output

```
proc print data=work.subset noobs;
run;
```

Partial PROC PRINT Output

First_ Name	Last_Name	Gender	Job_Title	Country	Employee_ ID	Salary
Tom	Zhou	M	Sales Manager	AU	120102	108255
Wilson	Dawes	M	Sales Manager	AU	120103	87975
Irenie	Elvish	F	Sales Rep. II	AU	120121	26600
Christina	Ngan	F	Sales Rep. II	AU	120122	27475
Kimiko	Hotstone	F	Sales Rep. I	AU	120123	26190

The character values are no longer truncated, but the order of the variables has changed.

46 p108d02

LENGTH Statement to Define the Length of Both Numeric and Character Variables





The LENGTH Statement

A LENGTH statement contains the names of the variables followed by the number of bytes to be used for storage.

```
LENGTH  RECTYPE  3 HHX  $ 6 INTV_QRT 3 INTV_MON 3;
```

The length of a variable is the number of bytes SAS allocates for storing the variable. It is not necessarily the same as the number of characters in the variable. The LENGTH statement is used in the data step to reduce the size (in terms of disk space) of SAS data sets.

By default, SAS uses 8 bytes to store numeric variables. Variables containing integer values can be stored using less than 8 bytes.

Largest integer represented exactly by length for SAS variables under Windows

Length in bytes	Largest integer represented exactly
3	8,192
4	2,097,152
5	536,870,912
6	137,438,953,472
7	35,184,372,088,832
8	9,007,199,254,740,990

Data sets containing many integer variables (such are common with data collected by questionnaire) or indicator variables can be reduced in size by more than 50%.

Variables containing real numbers should be left with the default length of 8. Note that specifying a length less than that required will result in a loss of precision without any warning being given.

GitHub: SASGateway/SASCourse/Week2/Ex23_Length.sas

Reading Non-Standard Data: Modifier with an Informat instead of the Informat Statement

Considerations

Use modified list input to read all the fields from **sales.csv**. Store the date fields as SAS dates.

Partial **sales.csv**

```
120102,Tom,Zhou,M,108255,Sales Manager,AU,11AUG1973,06/01/1993
120103,Wilson,Dawes,M,87975,Sales Manager,AU,22JAN1953,01/01/1978
120121,Irenie,Elvish,F,26600,Sales Rep. II,AU,02AUG1948,01/01/1978
120122,Christina,Ngan,F,27475,Sales Rep. II,AU,27JUL1958,07/01/1982
120123,Kimiko,Hotstone,F,26190,Sales Rep. I,AU,28SEP1968,10/01/1989
```

63

Modified List Input

This DATA step uses *modified list input.* Instead of a LENGTH statement, an informat specifies the length for each character variable.

input *variable* <:*informat.*> …;

```
data work.subset;
   infile "&path\sales.csv" dlm=',';
   input Employee_ID First_Name :$12.
         Last_Name :$18. Gender :$1. Salary
         Job_Title :$25. Country :$2.;
run;
```

- The **$12.** informat defines a length of 12 for **First_Name** and allows up to 12 characters to be read.
- The **:** format modifier tells SAS to read until it encounters a delimiter.

64 p108d05

INFILE Options

```
INFILE "raw-data-file" <DLM=> <DSD> <MISSOVER>;
```

Option	Description
DLM=	Specifies an alternate delimiter.
DSD	Sets the default delimiter to a comma, treats consecutive delimiters as missing values, and allows embedded delimiters when the data value is enclosed in quotation marks.
MISSOVER	Sets variables to missing if the end of the record is reached before finding values for all fields.

100

List Input (DLM= Option on the INFILE Statement)

DLM= Option

The DLM= option specifies a delimiter to be used for list input. Blank is the default delimiter.

```
Chloe,2,11/10/1995,$5,Running,Music,Gymnastics
Travis,2,1/30/1998,$2,Baseball,Nintendo,Reading
Jennifer,0,8/21/1999,$0,Soccer,Painting,Dancing
```

Input Raw Data File

```
data work.kids4;
   infile 'kids4.dat' dlm=',';
   input name $
         siblings
         bdate : mmddyy10.
         allowance : comma2.
         hobby1 : $10.
         hobby2 : $10.
         hobby3 : $10.;
run;
```

69

1. DSD Option with List Input





Missing Values at the End of a Record

The raw data file **phone.csv** contains missing values at the end of some records.

phone.csv

```
          1    1    2                    4    4
1---5----0----5----0----5----0----5----0----5
James Kvarniq,(704) 293-8126,(701) 281-8923
Sandrina Stephano,(919) 871-7830
Cornelia Krahl,(212) 891-3241,(212) 233-5413
Karen Ballinger,(714) 344-4321
Elke Wallstab,(910) 763-5561,(910) 545-3421
```

missing values

The DSD option is not appropriate because the missing data is not marked by consecutive delimiters.

97

MISSOVER Option

The *MISSOVER option* prevents SAS from loading a new record when the end of the current record is reached.

```
data contacts;
   length Name $ 20 Phone Mobile $ 14;
   infile "&path\phone.csv" dlm=',' missover;
   input Name $ Phone $ Mobile $;
run;

proc print data=contacts noobs;
run;
```

INFILE "*raw-data-file*" <DLM=> MISSOVER;

If SAS reaches the end of a record without finding values for all fields, variables without values are set to missing.

p108d10

98

Use MISSOVER if the last field or fields might be missing and you want SAS to assign missing values to the corresponding variable.

Viewing the Output

Partial SAS Log

```
NOTE: 5 records were read from the infile "S:\workshop\phone.csv".
      The minimum record length was 31.
      The maximum record length was 44.
NOTE: The data set WORK.CONTACTS has 5 observations and 3
variables.
```

PROC PRINT Output

Name	Phone	Mobile
James Kvarniq	(704) 293-8126	(701) 281-8923
Sandrina Stephano	(919) 871-7830	
Cornelia Krahl	(212) 891-3241	(212) 233-5413
Karen Ballinger	(714) 344-4321	
Elke Wallstab	(910) 763-5561	(910) 545-3421

99

GitHub: SASGateway/SASCourse/Week2 Ex8_List_Input_Modified_Input.sas
GitHub: SASGateway/SASCourse/Week2Ex9_DLM_DSD_MISSOVER.sas
GitHub: SASGateway/SASCourse/Week2Ex10_Modified_List_Input.sas
GitHub: SASGateway/SASCourse/Week2Ex24_Informat_List_Input_Formatted_Input.sas
GitHub: SASGateway\SASCourse\Ex27_Amper_Modifier.sas

Summary: Ampersand (&) and Colon (:) Format Modifiers

- The ampersand (&) format modifier enables you to read character values that contains one or more embedded blanks with list input and to specify a character informat. SAS reads until it encounters **two consecutive blanks**, the defined length of the variable, or the end of the input line, whichever comes first.
- The colon (:) format modifier enables you to use list input but also to specify an informat after a variable name, whether character or numeric. SAS reads until it encounters a blank column, the defined length of the variable (character only), or the end of the data line, whichever comes first.
- With modified list input, the INFORMAT specifies the length for each character variable, not the number of columns that are read.



GitHub: SASGateway/SASCourse/Week2/Ex30_@@.sas.

Acknowledgments: Portions of SAS' copyrighted SAS course content are reproduced here with permission of SAS Institute Inc., Cary, NC, USA. SAS Institute Inc. makes no warranties with respect to these materials and disclaims all liability therefor. Parts of additional materials are adapted from SAS Documentation, and web-based resources. Please let me know of any errors or typos you have found. Do not copy or circulate the handouts.

Multiple INPUT Statements

By default, SAS advances the pointer to column 1 of the next input record when SAS encounters an INPUT statement.

Input Raw Data File

```
Chloe    2
11/10/1995 $5
Running Music Gymnastics
Travis   2
1/30/1998  $2
Baseball Nintendo Reading
Jennifer 0
8/21/1999  $0
Soccer Painting Dancing
```

```
data work.kids8;
   infile 'kids8.dat';
   input name $ 1-8
         siblings 10;
   input @1 bdate mmddyy10.
         @12 allowance comma2.;
   input hobby1:$10.
         hobby2:$10.
         hobby3:$10.;
run;
```

82

Line-Pointer Controls

The / line-pointer control advances the pointer to column 1 of the next input record.

```
data work.kids8;
   infile 'kids8.dat';
   input name $ 1-8
         siblings 10
       / @1 bdate mmddyy10.
         @12 allowance comma2.
       / hobby1:$10.
         hobby2:$10.
         hobby3:$10.;
run;
```

86

Line-Pointer Controls

The #*n* line-pointer control advances the pointer to column 1 of record *n*.

```
data work.kids8;
   infile 'kids8.dat';
   input #1 name $ 1-8
            siblings 10
         #2 @1 bdate mmddyy10.
            @12 allowance comma2.
         #3 hobby1:$10.
            hobby2:$10.
            hobby3:$10.;
run;
```

87

Line-Pointer Controls

The second record is skipped in each iteration of the DATA step.

```
input name $ 1-8 siblings 10 //
      hobby1:$10. hobby2:$10. hobby3:$10.;
```

```
input #1 name $ 1-8 siblings 10
      #3 hobby1:$10. hobby2:$10. hobby3:$10.;
```

```
input name $ 1-8 siblings 10;
input;
input hobby1:$10. hobby2:$10. hobby3:$10.;
```

Output Data Set

VIEWTABLE: Work.Kids8

	name	siblings	hobby1	hobby2	hobby3
1	Chloe	2	Running	Music	Gymnastics
2	Travis	2	Baseball	Nintendo	Reading
3	Jennifer	0	Soccer	Painting	Dancing

90

6. Which one of the following INPUT statements correctly reads the values for **Fname** (in the second field), **Lname**, **Department**, and **Salary** (in that order)?

   a. input @14 Fname $10. @1 Lname $10./
      Department $7./
      Salary comma10.;
   b. input @14 Fname $10.
      @1 Lname $10.
      #2 Department $7.
      #3 Salary comma10.;
   c. both a and b

```
          1    1    2    2
1---5----0----5----0----5
ABRAMS         THOMAS
SALES
$45,209.03
BARCLAY        ROBERT
ACCTING
$49,180.36
COURTNEY        MARK
EDUC
$44,006.16
```

You can use the @*n* column pointer control to read the values for Lname and Fname. You can use either the / or #*n* line pointer control to advance the input pointer to the second and third lines of the raw data file to read the values for Department and Salary.

94

When to Use the Forward Slash (/) Line Pointer Control

You have data elements that need to be read sequentially from multiple records to create a single observation.

GitHub:
SASGateway/SASCourse/Week2Ex13_Line_Pointer_controls.sas

GitHub: SASGateway/SASCourse/Week2Ex11_Question_marks.sas
GitHub: SASGateway/SASCourse/Week2Ex12_Dumping_Records.sas

GitHub: SASGateway/SASCourse/Week2Ex17_Filename_Libname.sas

Example: Mix of Column Input, Formatted Input, and List Input Styles

INPUT Statement

```
----|----10---|----20---|----30---|----40---|----50
Chloe    2 11/10/1995 $5Running Music Gymnastics
Travis   2 1/30/1998  $2Baseball Nintendo Reading
Jennifer 0 8/21/1999  $0Soccer Painting Dancing
```

Input Raw Data File

The following three ways can describe a record's values in the INPUT statement:

- column input
- formatted input
- list input

```
input name $ 1-8 siblings 10
      @12 bdate mmddyy10.
      @23 allowance comma2.
      hobby1 $ hobby2 $ hobby3 $;
```

column

formatted

list

11

Column Input

With column input, the column numbers that contain the value follow a variable name in the INPUT statement.

```
input name $ 1-8 siblings 10
      @12 bdate mmddyy10.
      @23 allowance comma2.
      hobby1 $ hobby2 $ hobby3 $;
```

To read with column input, data values

- must be in the same columns in all the input data records
- must be in standard form.

12

Formatted Input

With formatted input, an informat follows a variable name and defines how SAS reads the values of this variable. An informat gives the data type and the field width of an input value.

```
input name $ 1-8 siblings 10
      @12 bdate mmddyy10.
      @23 allowance comma2.
      hobby1 $ hobby2 $ hobby3 $;
```

To read with formatted input, data values

- must be in the same columns in all the input data records
- can be in standard or nonstandard form.

13

List Input

With list input, variable names in the INPUT statement are specified in the same order that the fields appear in the input data records.

```
input name $ 1-8 siblings 10
      @12 bdate mmddyy10.
      @23 allowance comma2.
      hobby1 $ hobby2 $ hobby3 $;
```

To read with list input, data values

- must be separated with a delimiter
- can be in standard or nonstandard form.

14

The Use of Multiple Input Statements

```
 1 *Ex29_Multiple_Input_Statements.sas;
 2 data work.HAVE(drop=i);
 3  input date: Anydtdte9. @;
 4  do i = 1 to 4;
 5  input name $ study_hours @;
 6  output;
 7  end;
 8 datalines;
 9 27Aug2018 Doris 5.5 Alice 4.0 Mike 2.0 James 1.0
10 28Jun2018 Doris 3.0 Alice 3.0 Mike 3.0 James 1.0
11 ;
12 proc print data=work.HAVE;
13 Format date worddate.;
14 run;
```

Obs	date	name	study_hours
1	August 27, 2018	Doris	5.5
2	August 27, 2018	Alice	4.0
3	August 27, 2018	Mike	2.0
4	August 27, 2018	James	1.0
5	June 28, 2018	Doris	3.0
6	June 28, 2018	Alice	3.0
7	June 28, 2018	Mike	3.0
8	June 28, 2018	James	1.0

Normally, each INPUT statement in a DATA step reads a new data record into the input buffer. However, when you use a trailing @, the following things occur:

- The pointer position does not change.
- No new record is read into the input buffer.
- The next INPUT statement for the same iteration of the DATA step continues to read the same record rather than a new one.

SAS releases a record held by a trailing @ when

- a null INPUT statement executes (**input;**)
- an INPUT statement without a trailing @ executes
- the next iteration of the DATA step begins.

Reading Data from Single/Multiple Records (@@ at the End of the Input Statement)

```
 1 *Ex30_@@.sas;
 2 data work.HAVE;
 3  input date: Anydtdte. name $ study_hours @@;
 4 datalines;
 5 27Aug2018 Doris 5.5 28Aug2018 Alice 4.0
 6 29Aug2018 Mike 2.0 29Aug2018 James 1.0
 7 30Jun2018 Doris 3.0 31Aug2018 Alice 3.0
 8 01Sep2018 Mike 3.0
 9 02Sep2018 James 1.0
10 ;
11 proc print data=work.HAVE;
12 Format date mmddyy10.;
13 run;
```

In the above program, we are reading 5 lines of data with the **Datalines** statement. Lines 5, 6, and 7 each have have two records; and lines 8 and 9 each have one record. Here, our goal is to create a SAS data file that will have one observation from each record – a total of 8 observations.

Notice that the Informat for the variable **date** does not specify a *w* value because we are reading data using List Input with a colon modifier; here, SAS reads data fields until it encounters a blank.

a blank column

Obs	date	name	study_ hours
1	08/27/2018	Doris	5.5
2	08/28/2018	Alice	4.0
3	08/29/2018	Mike	2.0
4	08/29/2018	James	1.0
5	06/30/2018	Doris	3.0
6	08/31/2018	Alice	3.0
7	09/01/2018	Mike	3.0
8	09/02/2018	James	1.0

How to Read Multiple Data Files

3 data files (testfile1.csv, testfile2.csv, and testfile3.csv) shown below:

A,123,345	D,456,334	G,456,334
B,456,456	E,456,223	H,456,223
C,789,678	F,876,456	I,876,456

GitHub: SASGateway/SASCourse/WeekEx28_Reading_Multiple_Files.sas

Placement of the Subsetting IF Statement

Generally, the most efficient place to put the subsetting IF statement is as soon as all the variables that are needed to evaluate the condition are assigned values.

```
data EuropeQ1;
   infile "&path\sales.dat";
   input @6 Location $3. @;
   if Location = 'EUR';
   input  @1 SaleID $4.
          @10 SaleDate date9.
          @20 Amount commax7.;
run;
```

78

p204d07

SAS releases a record held by a trailing @ when the next iteration of the DATA step begins.

Subsetting Mixed Record Types: Output

```
proc print data=EuropeQ1 noobs;
   var SaleID Location SaleDate Amount;
run;
```

PROC PRINT Output

Sale ID	Location	Sale Date	Amount
3034	EUR	18657	1876.3
1345	EUR	18664	3145.6

79

```
1  *Ex25_read_from_web.sas;
2  Filename raw url
3      'http://data.princeton.edu/wws509/datasets/effort.dat';
4  data have1;
5     infile raw  firstobs=2 truncover ;
6     input record $80. ;
7     put _all_;
8     if _n_=5 then stop;
9  run;
10 proc print data=have1; run;
11
12 data have2;
13    infile raw  firstobs=2 obs=5 truncover ;
14    input country $ setting  effort  change ;
15    put _all_;
16  run;
17 proc print data=have2; run;
```

```
1  *Ex26_Modified_List_Input;
2  Options ls=132;
3  data annual_exp2013;
4  infile datalines;
5  input Age  (Food Housing Clothing Transportation Healthcare
6        Entertainment Pension_S Other)(:comma.);
7
8  FORMAT Food Housing Clothing Transportation Healthcare
9        Entertainment Pension_S Other dollar10.;
10 datalines;
11 0  4,698    10,379  1,513   5,672     943   1,243   2,153   3,771
12 25 6,197    17,207  1,832   9,183   2,189   2,214   5,178   4,087
13 35 7,920    20,619  1,960   10,519  3,188   2,958   6,791   4,827
14 45 7,907    19,001  1,826   10,782  3,801   3,070   7,305   6,833
15 55 6,711    17,937  1,563   9,482   4,378   2,651   6,593   6,577
16 65 6,020    15,639  1,222   7,972   5,188   2,488   2,833   5,394
17 75 4,144    12,314  768     5,149   4,910   1,422     832   4,844
18 ;
19 run;
20 proc print noobs; run;
```

@@ at the End of the INPUT Statement



When to Use Named Input

- Data records contain both variable names and values

What to Specify on the Input Statement (Named Input)

- Each variable name must be followed by an equals on the INPUT statement
- Each variable name must be followed by an equals on the data as well
- Character variables must be indicated by a "$" following the equals sign on the INPUT statement
- Appropriate format modifiers and informats must be specified
- The"/" at the end of the line must be used to read the next data line in order to complete the observation

```
1  *Example_Named_Input.sas;
2  options nocenter nodate nonumber ls=132;
3  DATA TEST;
4  input name = & $ 30. address = & $ 30.
5        city_zip  = & $ 30. phone= $ 14.
6        Num_employees = ;
7        FORMAT Num_employees comma7.;
8  DATALINES;
9  name=Air Force Personnel Center /
10 address=550 C Street West /
11 city_zip=Randolph AFB, TX 78150 /
12 phone=1-800-525-0102 /
13 Num_employees=5876
14 name= Navy Personnel Command /
15 address= 5720 Integrity Drive /
16 city_zip= Millington, TN 38055 /
17 phone= 901-874-4885 /
18 Num_employees=3987
19 ;
20 proc print noobs; run;
```

name	address	city_zip	phone	Num_ employees
Air Force Personnel Center	550 C Street West	Randolph AFB, TX 78150	1-800-525-0102	5,876
Navy Personnel Command	5720 Integrity Drive	Millington, TN 38055	901-874-4885	3,987

Reading Data from Zipped Files

You need to download the zip file (national data) from this site in order to run the SAS code bellow.

https://www.ssa.gov/oact/babynames/limits.html

```
*Ex18_Read_Zipped_File2.sas;
Filename ZIPFILE SASZIPAM 'c:\SASCourse\Week2\names.zip';
DATA newdata;
  INFILE ZIPFILE(yob1920.txt) DLM=',';
  INPUT gender $ name $ number;
RUN;
proc print data=newdata (obs=5); run;
```

The FILENAME statement specifies the type of file that needs to be unzipped (i.e. zipfile).

The engine SASZIPAM is used to decompress the file.

GitHub:
SASGateway/SASCourse/Week2/Ex18_Read_Zipped_Files2.sas

Some Common Ways to Read Excel Workbooks into SAS

- SAS Import Wizard
- PROC IMPORT using the DBMS=XLS option
- PROC IMPORT using the DBMS=XLSX option
- LIBNAME Engine for Excel

Example: Reading an Excel Workbook into SAS - PROC IMPORT Using the DBMS=xlsx option (SAS Code)

```
options nodate nonumber nodate;
PROC IMPORT DATAFILE= "C:\SASCourse\Week2\class.xlsx"
    dbms=xlsx REPLACE OUT= work.class_x;
    SHEET="Sheet1";
    GETNAMES=YES;
RUN;
PROC PRINT data=work.class_x;
RUN;
```

GitHub: SASGateway/SASCourse/Week2/Ex20_import_Excel_x.sas

SAS/ACCESS LIBNAME Statement

The default Excel engine can be specified when the bit count of SAS and Microsoft Office are the same (both are 32-bit or both are 64-bit).

```
libname myxls excel 'customers.xls';
```

The PC Files Server engine must be specified when the bit counts differ.

```
libname myxls pcfiles path='customers.xls';
```

110

XLS Files

Prior to Office 2007, XLS was the default format for Excel to store data in worksheets, charts, and macros.

The XLS format is

- a proprietary binary file format called *Binary Interchange File Format* (BIFF)
- readable by all Microsoft Excel versions.

35

XLSX Files

With the release of Microsoft Office 2007, Microsoft changed the default file format. To facilitate this new format, Microsoft added an additional X to their document extensions. For Excel, this is XLSX.

The XLSX file format

- is readable only by versions 2007 and later
- does not support Excel macros.

36

Excel File Structure Limitation

File Type	Row Limit	Column Limit
XLS	65,536	255
XLSX	1,048,576	16,384

38

Reading an Excel Workbook: LIBNAME Engine for Excel

- The XLSX engine accesses the XLSX file directly when reading the Excel data into SAS
- Bitness (32-bit versus 64-bit) does not matter

The LIBNAME statement references the whole Excel file, which is viewed as a SAS library and, the members inside (spreadsheet or named range) are viewed as data files. The SET statement uses the Excel sheet as an input data file for this data step. Below is the SAS Code.

```
options validvarname=any;
libname XL XLSX 'C:\SASCourse\Week2\class.xlsx';
data work.class;
  set XL.Sheet1;
run;
libname XL CLEAR;
```

The setting of the VALIDVARNAME system option allows the use of column names that contain embedded spaces and special characters.

the first LIBNAME statement references the whole Excel file, which is viewed as a SAS library and, the member inside (spread sheet or named range) is viewed as a data file.

The SET statement uses the Excel sheet as an input data file for this data step.

The last LIBNAME specifies the libref and clear option to disassociate the libref from the SAS library.



In addition to the above two ordinary missing values, there are 26 special missing values that only apply to the numeric variable.

MISSING= System Option

The MISSING= system option allows to specify a character to be printed when numeric variables contain ordinary missing values (.).

MISSING Statement

If your data contain characters that represent special missing values, such as a or z, do not use the MISSING= system option to define them; simply define these values in a MISSING statement.

`GitHub: SASGateWay/SASCourse/Week2/Ex19_Missing.sas`

DATA Step Processing



22

Compilation Phase

During compilation, SAS does the following:

- scans the step for syntax errors
- translates each statement into machine language
- creates an *input buffer* to hold one record at a time from the raw data file



- creates the program data vector (PDV) to hold one observation
- creates the descriptor portion of the output data set

23

Compilation: Formatted Input

```
data work.discounts;
   infile "&path\offers.dat";
   input @1 Cust_type 4.
         @5 Offer_dt mmddyy8.
         @14 Item_gp $8.
         @22 Discount percent3.;
run;
```

Input Buffer



PDV

Cust_type N 8	Offer_dt N 8	Item_gp $ 8	Discount N 8

19 ...

Execution: Formatted Input

```
data work.discounts;
   infile "&path\offers.dat";
   input @1 Cust_type 4.
         @5 Offer_dt mmddyy8.
         @14 Item_gp $8.
         @22 Discount percent3.;
run;
```

Initialize PDV

Input Buffer



PDV

Cust_type N 8	Offer_dt N 8	Item_gp $ 8	Discount N 8
.	.		.

20 ...

Execution: Formatted Input

Specify input data file

```
data work.discounts;
   infile "&path\offers.dat";
   input @1 Cust_type 4.
         @5 Offer_dt mmddyy8.
         @14 Item_gp $8.
         @22 Discount percent3.;
run;
```

Input Buffer

```
          1                   2
1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 0 1 2 3 4 5
```

PDV

Cust_type N 8	Offer_dt N 8	Item_gp $ 8	Discount N 8
.	.		.

21 ...

Execution: Formatted Input

Load input buffer

```
data work.discounts;
   infile "&path\offers.dat";
   input @1 Cust_type 4.
         @5 Offer_dt mmddyy8.
         @14 Item_gp $8.
         @22 Discount percent3.;
run;
```

Input Buffer

```
          1                   2
1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 0 1 2 3 4 5
1 0 4 0 1 2 / 0 2 / 1 1   O u t d o o r s 1 5 %
```

PDV

Cust_type N 8	Offer_dt N 8	Item_gp $ 8	Discount N 8
.	.		.

22 ...

Execution: Formatted Input

```
data work.discounts;
   infile "&path\offers.dat";
   input @1 Cust_type 4.
         @5 Offer_dt mmddyy8.
         @14 Item_gp $8.
         @22 Discount percent3.;
run;
```

Load first value into PDV

Input Buffer

```
          1                   2
1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 0 1 2 3 4 5
1 0 4 0 1 2 / 0 2 / 1 1   O u t d o o r s 1 5 %
```

PDV

Cust_type N 8	Offer_dt N 8	Item_gp $ 8	Discount N 8
1040	.		.

23 ...

Execution: Formatted Input

```
data work.discounts;
   infile "&path\offers.dat";
   input @1 Cust_type 4.
         @5 Offer_dt mmddyy8.
         @14 Item_gp $8.
         @22 Discount percent3.;
run;
```

Load second value into PDV

Input Buffer

```
          1                   2
1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 0 1 2 3 4 5
1 0 4 0 1 2 / 0 2 / 1 1   O u t d o o r s 1 5 %
```

PDV

Cust_type N 8	Offer_dt N 8	Item_gp $ 8	Discount N 8
1040	18963		.

24 ...

Execution: Formatted Input

```
data work.discounts;
   infile "&path\offers.dat";
   input @1 Cust_type 4.
         @5 Offer_dt mmddyy8.
         @14 Item_gp $8.
         @22 Discount percent3.;
run;
```

Load third value into PDV

Input Buffer

```
          1                   2
1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 0 1 2 3 4 5
1 0 4 0 1 2 / 0 2 / 1 1   O u t d o o r s 1 5 %
```

PDV

Cust_type N 8	Offer_dt N 8	Item_gp $ 8	Discount N 8
1040	18963	Outdoors	.

25 ...

Execution: Formatted Input

```
data work.discounts;
   infile "&path\offers.dat";
   input @1 Cust_type 4.
         @5 Offer_dt mmddyy8.
         @14 Item_gp $8.
         @22 Discount percent3.;
run;
```

Load fourth value into PDV

Input Buffer

```
          1                   2
1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 0 1 2 3 4 5
1 0 4 0 1 2 / 0 2 / 1 1   O u t d o o r s 1 5 %
```

PDV

Cust_type N 8	Offer_dt N 8	Item_gp $ 8	Discount N 8
1040	18963	Outdoors	.15

26 ...

Execution: Formatted Input

```
data work.discounts;
   infile "&path\offers.dat";
   input @1 Cust_type 4.
         @5 Offer_dt mmddyy8.
         @14 Item_gp $8.
         @22 Discount percent3.;
run;
```

Implicit OUTPUT;
Implicit RETURN;

Input Buffer

```
          1         2
1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 0 1 2 3 4 5
1 0 4 0 1 2 / 0 2 / 1 1   O u t d o o r s 1 5 %
```

PDV

Cust_type N 8	Offer_dt N 8	Item_gp $ 8	Discount N 8
1040	18963	Outdoors	.15

27 ...

Execution: Formatted Input

Reinitialize PDV

```
data work.discounts;
   infile "&path\offers.dat";
   input @1 Cust_type 4.
         @5 Offer_dt mmddyy8.
         @14 Item_gp $8.
         @22 Discount percent3.;
run;
```

Continue until EOF

Input Buffer

```
          1         2
1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 0 1 2 3 4 5
1 0 4 0 1 2 / 0 2 / 1 1   O u t d o o r s 1 5 %
```

PDV

Cust_type N 8	Offer_dt N 8	Item_gp $ 8	Discount N 8
.	.		.

28













Acknowledgments: Portions of SAS' copyrighted SAS course content are reproduced here with permission of SAS Institute Inc., Cary, NC, USA. SAS Institute Inc. makes no warranties with respect to these materials and disclaims all liability therefor. Parts of additional materials are adapted from SAS Documentation, and web-based resources. Please let me know of any errors or typos you have found. Do not copy or circulate the handouts.





Execution

Partial **sales.csv**

```
120102,Tom,Zhou, ...
120103,Wilson,Dawes, ...
120121,Irenie,Elvish, ...
120122,Christina,Ngan, ...
120123,Kimiko,Hotstone, ...
120124,Lucian,Daymond, ...
120125,Fong,Hofmeister, ...
```

```
data work.subset;
  infile "&path\sales.csv"
         dlm=',';
  input Employee_ID First_Name $
        Last_Name $ Gender $
        Salary Job_Title $
        Country $;
run;
```

Implicit OUTPUT;
Implicit RETURN;

Input Buffer

```
1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 0 1 2 3 4 5
1 2 0 1 0 2 , T o m , Z h o u , M , 1 0 8 2 5 5 ,
```

PDV

Employee _ID N 8	First _ Name $ 8	Last _Name $ 8	Gender $ 8	Salary N 8	Job _ Title $ 8	Country $ 8
120102	Tom	Zhou	M	108255	Sales Ma	AU

31 ...

Execution

Here is the output data set after the first iteration of the DATA step.

work.subset

Employee _ID	First _Name	Last _Name	Gender	Salary	Job _Title	Country
120102	Tom	Zhou	M	108255	Sales Ma	AU

32 ...









Execution

Partial **sales.csv**

```
120102,Tom,Zhou, ...
120103,Wilson,Dawes, ...
120121,Irenie,Elvish, ...
120122,Christina,Ngan, ...
120123,Kimiko,Hotstone, ...
120124,Lucian,Daymond, ...
120125,Fong,Hofmeister, ...
```

```
data work.subset;
  infile "&path\sales.csv"
         dlm=',';
  input Employee_ID First_Name $
        Last_Name $ Gender $
        Salary Job_Title $
        Country $;
run;
```

Input Buffer

1	2	3	4	5	6	7	8	9	0	1	2	3	4	5	6	7	8	9	0	1	2	3	4	5
1	2	0	1	0	3	,	W	i	l	s	o	n	,	D	a	w	e	s	,	M	,	8	7	9

PDV

Employee _ID N 8	First _ Name $ 8	Last _Name $ 8	Gender $ 8	Salary N 8	Job _ Title $ 8	Country $ 8
120103	Wilson	Dawes	M	87975	Sales Ma	AU

37 ...

Execution

Partial **sales.csv**

```
120102,Tom,Zhou, ...
120103,Wilson,Dawes, ...
120121,Irenie,Elvish, ...
120122,Christina,Ngan, ...
120123,Kimiko,Hotstone, ...
120124,Lucian,Daymond, ...
120125,Fong,Hofmeister, ...
```

```
data work.subset;
  infile "&path\sales.csv"
         dlm=',';
  input Employee_ID First_Name $
        Last_Name $ Gender $
        Salary Job_Title $
        Country $;
run;
```

Implicit OUTPUT;
Implicit RETURN;

Input Buffer

1	2	3	4	5	6	7	8	9	0	1	2	3	4	5	6	7	8	9	0	1	2	3	4	5
1	2	0	1	0	3	,	W	i	l	s	o	n	,	D	a	w	e	s	,	M	,	8	7	9

work.subset

Employee _ID	First _Name	Last _Name	Gender	Salary	Job _Title	County
120102	Tom	Zhou	M	108255	Sales Ma	AU
120103	Wilson	Dawes	M	87975	Sales Ma	AU

38 ...



PUT and PUTLOG Statement

The PUT statement writes to the LOG or to an External File with a FILE statement, but the PUTLOG statement always writes to the LOG.

```
1  *Ex21_put.sas;
2  * List all DATA step variables and their values;
3  data _null_;
4     set sashelp.class(obs=2);
5     put _all_;
6  run;
```

Line 3: The keyword _NULL_ on the DATA statement is used to execute the data step without creating a data set.

Line 5: The PUT statement is used to create output records to the LOG window. The special SAS name list _ALL_ refers to all variables on the data step and program data vector including _N_ and _ERROR_.

```
Name=Alfred Sex=M Age=14 Height=69 Weight=112.5 _ERROR_=0 _N_=1
Name=Alice Sex=F Age=13 Height=56.5 Weight=84 _ERROR_=0 _N_=2
NOTE: There were 2 observations read from the data set SASHELP.CLASS.
```

```
 7  * Exclude automatic variables;
 8  data _null_;
 9     set sashelp.class(obs=2);
10     put (_all_)(=);
11  run;
```

Line 21: In the PUT statement, the variable list argument is _ALL_ [i.e., all variables on the data step and program data vector excluding _N_ and _ERROR_]; the format argument is the equal sign so that the variable name precedes its value.

```
Name=Alfred Sex=M Age=14 Height=69 Weight=112.5
Name=Alice Sex=F Age=13 Height=56.5 Weight=84
NOTE: There were 2 observations read from the data set SASHELP.CLASS.
```

```
17 /* Put each value on a new line and apply
18 a common format to all numeric variables*/
19 data _null_;
20    set sashelp.class(obs=2);
21    put (_all_)(=/12.2);
22 run;
```

Line 21: The PUT statement is used to tell SAS to write each of the variables in the data step and program data vector (except _N_ and _ERROR_) that must precede its value. An additional format argument / is used so that each variable and its value separated by an equal sign is output to a separate line. Note the **FORMAT***w.d* (i.e., 12.2) that applies to all numeric variables.

```
Name=Alfred
Sex=M
Age=14.00
Height=69.00
Weight=112.50

Name=Alice
Sex=F
Age=13.00
Height=56.50
Weight=84.00
NOTE: There were 2 observations read from the data set SASHELP.CLASS.
```

```
23 /* List values as a table and apply formats
24 to groups of variables*/
25 data _null_;
26   set sashelp.class(obs=2);
27   if _n_=1 then put @1 'NAME' @19 'SEX' @23 'AGE'
28                     @30 'HEIGHT' @38 'WEIGHT';
29   put (_all_)(1*$20.,1*$2.,1*3.,2*8.2);
30 run;
```

Line 27: During the first iteration, the variable names are printed each starting with the column position specified, to the LOG window by default.

Line 29: This is an example of the FORMATTED PUT statement. There is no equal sign or slash as a format-list argument. Formats specified as format-list arguments. For example, a character format **$20.** is applied to the variable NAME. Another character format **$2.** is applied to the variable SEX. A numeric format **3.** is applied to the variable AGE. Another numeric format 8.2 is applied to the variables HEIGHT and WEIGHT.

```
NAME                 SEX AGE    HEIGHT  WEIGHT
Alfred                 M  14    69.00  112.50
Alice                  F  13    56.50   84.00
NOTE: There were 2 observations read from the data set SASHELP.CLASS.
```

```
33 /* List values as a table and apply formats
34 to groups of variables. Route output to the
35 standard SAS output window. */
36 options nodate nonumber;
37 title;
38 data _null_;
39   set sashelp.class(obs=2);
40   file print;
41   if _n_=1 then put @1 'NAME' @19 'SEX' @23 'AGE'
42                     @30 'HEIGHT' @38 'WEIGHT';
43   put (_all_)(1*$20.,1*$2.,1*3.,2*8.2);
44 run;
```

Line 40: This statement creates the tabular output to the OUTPUT window, not the LOG window.

(Partial SAS Log)

```
NOTE: 3 lines were written to file PRINT.
```

Snapshot from the SAS output window

```
NAME                 SEX AGE    HEIGHT  WEIGHT
Alfred                 M  14   69.00  112.50
Alice                  F  13   56.50   84.00
```

```
46 /* List values as a table and apply formats
47 to groups of variables. Route output to a file */
48 options nodate nonumber;
49 title;
50 data _null_;
51   set sashelp.class(obs=2);
52   file "E:\class_data.txt";
53   if _n_=1 then put @1 'NAME' @19 'SEX' @23 'AGE'
54                     @30 'HEIGHT' @38 'WEIGHT';
55   put (_all_)(1*$20.,1*$2.,1*3.,2*8.2);
56 run;
```

Line 52: This statement creates a tabular raw data file named CLASS_DATA.TXT in the path specified, all enclosed in quotation mark.

```
NOTE: The file "E:\class_data.txt" is:
      Filename=E:\class_data.txt,
      RECFM=V,LRECL=256,File Size (bytes)=0,
      Last Modified=20Aug2016:10:44:00,
      Create Time=20Aug2016:10:42:55

NOTE: 3 records were written to the file "E:\class_data.txt".
      The minimum record length was 41.
      The maximum record length was 43.
NOTE: There were 2 observations read from the data set SASHELP.CLASS.
```

```
58 /* List values as a table and apply formats
59 to groups of variables. Route output to a file
60 Write an informational message to the log using
61 an PUTLOG statement */
62 options nodate nonumber;
63 title;
64 data _null_;
65   set sashelp.class(obs=2)END=last;
66   file "E:\class_data2.txt";
67   if _n_=1 then put @1 'NAME' @19 'SEX' @23 'AGE'
68                     @30 'HEIGHT' @38 'WEIGHT';
69   put (_all_)(1*$20.,1*$2.,1*3.,2*8.2);
70   if last then putlog "User's NOTE: Writing to the File is completed";
71 run;
```

```
NOTE: The file "E:\class_data2.txt" is:
      Filename=E:\class_data2.txt,
      RECFM=V,LRECL=256,File Size (bytes)=0,
      Last Modified=20Aug2016:11:05:30,
      Create Time=20Aug2016:11:04:06

User's NOTE: Writing to the File is completed
NOTE: 3 records were written to the file "E:\class_data2.txt".
```

Since the FILE statement in line 66 is in effect, the PUT statement in line 69 writes to the external File (class_data2.txt).

The PUTLOG statement in line 70 is used to write an informational message to the LOG. Note that we have preceded a message text with User's Note to better identify the output in the log.

In line 65, "the END= option defines a temporary *variable* whose value is 1 when the DATA step is processing the last observation. At all other times, the value of *variable* is 0. Although the DATA step can use the END= variable, SAS does not add it to the resulting data set". [SAS Documentation]

GitHub: SASGateWay/SASCourse/Week2/Ex21_put.sas

GitHub: SASGateWay/SASCourse/Week2/Ex22__DM_CSV.sas